Writing Control Software for the HP 83236A/ B

Product Note

For testing PCS mobiles and base stations with the HP 8920A/ B, 8921A, or 8924C, test set
Table of Contents

Using the HP 83236 as an RF source ............... 4
Using the HP 83236 as an RF analyzer .......... 11
Making accurate power measurements ....... 15
Introduction

The HP 83236A/B is a frequency translator and power meter that is intended to be used with an HP 8920A/B, 8921B, or 8924C Test Set for testing PCS mobiles and base stations.

This programming note is written to help programmers understand the concepts and algorithms necessary for programming the HP 83236A/B. This note is applicable to programmers that are directly controlling the HP 83236A/B with HP-IB control. For additional information, including programming, consult the HP 83236A/B operating manual (83236-90001).

The HP 83236A contains only HP-IB control, where the HP 83236B contains both HP-IB and a special serial control directly from the test set. For programmers using the HP 83236A, it is necessary to use the HP-IB control methods described in this document. For programmers using the HP 83236B, there are two choices, either HP-IB or direct serial control.

The highly recommended method for using a PCS interface with an HP 892X Test Set is to use the HP 83236B (serial control technique) with an HP 892X Test Set that has special firmware supporting the serial control method. Unfortunately, the special firmware for the test sets is not currently available and anyone using the HP 83236A or B will need to use the HP-IB control method described in this document (at least until firmware is available).

Each section of this document contains an explanation of the programming functions and a flowchart showing the program flow and HP-IB commands. A sample program is also included that shows simple HP BASIC routines for setting most of the HP-IB functions.

Your feedback on this document is important to Hewlett-Packard. Please e-mail or FAX your comments to:

Allen Henley
e-mail: henley@spk.hp.com
FAX: 1-509-921-3700
Using the HP 83236 as an RF Source

The HP 83236 contains a programmable oscillator and programmable attenuator that is used for upconverting the RF output signal from the HP 892X. When using the HP 83236 for this application, the programmer needs to consider calibration, setting frequency, and setting level. The following discussions explain some of the factors to consider, and the HP-IB techniques for programming each function. Refer to the graphic on page 6 for a visual representation of the algorithm for correctly setting the RF frequency and RF level.

Setting RF Frequency
To set the RF output frequency of the HP 83236, it is necessary to perform a two step process. The first step is to set and query the HP 83236, the second step is to set the test set output frequency.

The HP-IB commands for the HP 83236 to perform the first step would be something like:

```
OUTPUT 719; "RX:TSET:FREQ? 1900 MHZ"
ENTER 719; Hp892x_freq
```

The OUTPUT line instructs the HP 83236 to set its internal oscillator to the correct frequency, and the ENTER statement reads frequency data from the HP 83236 so the control program will know the correct frequency to program the test set.

To perform the second step use the following command (to the test set):

```
OUTPUT 714; "RFG:FREQ "; Hp892x_freq
```

This will program the test set to the frequency which was specified by the HP 83236.

It is important to remember that when setting RF frequency and RF level, it is necessary to set the frequency first, then set the RF level. In situations where only the RF frequency is changing, it is necessary to set the frequency, then to re-set the level. This is because the HP 83236 may require a new level from the test set for the new RF frequency.
Setting RF Level

Always set the RF frequency before setting the RF level. The HP 83236 uses calibration factors that are based on RF frequency.

The steps necessary for setting RF output level are similar to the steps for setting RF frequency. The procedure is a two step process that first sets and queries the HP 83236, then sets the test set to the correct output level.

The HP-IB commands to the HP 83236 to perform the first step would be something like:

```
OUTPUT 719; "RX:TSET:LEV? -20 DBM"
ENTER 719; Hp892x_level
```

The OUTPUT line instructs the HP 83236 to set its internal attenuators based on frequency and level, and the ENTER statement reads data back from the HP 83236 so the control program will know the correct level for the test set.

To perform the second step use the following command (to the test set):

```
OUTPUT 714; "RFG:AMPL " ; Hp892x_level ; "DBM"
```

This will program the test set to the level which was specified by the HP 83236.

Since it is necessary to set the level each time frequency or level is changed, it may be convenient to check temperature calibration as part of the control program subroutine that sets RF level. Notice in the example program that each time level is set in the subroutine "Set_output_lvl" another subroutine is called "Check_gen_cal" which verifies the temperature calibration. In a real application, the control program would most likely want to verify that if a mobile call was connected and the temperature calibration indicated it needed to be run, the mobile call would be properly terminated before the calibration routine was run.
Setting the HP 83236 as an RF Generator

START

Check Temperature Cal

OUTPUT Hp83236; "COMP:TEMP:REQ:STAT?"
ENTER Hp83236; Gen_cal_req

Calibration Needed
Calibrate Necessary Bands

Calibration Not Needed

Set and Query RF Frequency on HP 83236

OUTPUT Hp83236; "RX:TSET:FREQ? 1930 MHz"
ENTER Hp83236; Hp892x_freq

Set RF Frequency on HP 892X

OUTPUT Hp892x; "RFG:FREQ "; Hp892x_freq

Set and Query RF Level on HP 83236

OUTPUT Hp83236; "RX:TSET:LEV? -20 dBm"
ENTER Hp83236; Hp892x_level

Set RF Level on HP 892X

OUTPUT Hp892x; "RFG:AMPL "; Hp892x_level; "dbm"

DONE
Calibration

To maintain guaranteed specifications, it is necessary to check the RF generator temperature calibration frequently. For specified performance, the generator should be recalibrated whenever there is a +/- 1 degree C change in temperature. The HP 83236 has a built-in temperature sensor that detects changes in temperature, and it can easily be queried to let the program know if a calibration is required. To query the sensor use the HP-IB query "COMP:TEMP:REQ:STAT?". This query will return a 1 if compensation is required, it will return a 0 if compensation is not required.

Because calibration takes about five seconds per frequency band, and most applications will only use one or two bands, the calibration routine is generally only executed for the frequency bands of interest. The flowchart and example program shows a general purpose calibration routine which calibrates bands zero to four. In your application, you should only calibrate the bands you intend to use. See the subroutine "Tcal_gen_lvl" in the example program to see the HP-IB commands necessary for the calibration. Because the calibration requires setting levels and frequencies on the test set, it is not possible to run this calibration while a mobile call is connected. The control program should check calibration before the mobile call is established.

Note: New firmware is being added that allows the control program to query each frequency band individually to determine if calibration is required for that specific band. Refer to the latest documentation for your product to determine if this feature is available.

Frequency Bands:

<table>
<thead>
<tr>
<th>Band</th>
<th>Frequencies (MHz) HP 83236A</th>
<th>Frequencies (MHz) HP 83236B</th>
</tr>
</thead>
<tbody>
<tr>
<td>0</td>
<td>824 to &lt;859</td>
<td>824-859</td>
</tr>
<tr>
<td>1</td>
<td>859 to 894</td>
<td>859 to 894</td>
</tr>
<tr>
<td>2</td>
<td>1850 to 1910</td>
<td>1710 to &lt;1785</td>
</tr>
<tr>
<td>3</td>
<td>1910 to &lt;1930</td>
<td>1805 to &lt;1880</td>
</tr>
<tr>
<td>4</td>
<td>1930 to 1990</td>
<td>1850 to &lt;1910</td>
</tr>
<tr>
<td>5</td>
<td>None</td>
<td>1930 to &lt;1990</td>
</tr>
</tbody>
</table>
Temperature Calibration for the HP 83236 RF Generator

START

FOR Testband = 0 to 4

Query HP 83236 for Compensation Freq

Query HP 83236 for Compensation Level

Set HP 892X Level and Frequency

WAIT 1

Execute and Query HP 83236 Temperature Compensation

Check Done_flag

Note: If calibration is successful then Done_flag = 1, else there was an error during calibration.

NEXT Testband
HP BASIC Example Program for RF Source

10 ! re-save "c:\hpbasic\diet\generate.pgm"
20 ! Rev. A.02.00
30 ! Example of using the HP 83236 as an upconverter/
40 ! RF Source for use with the HP 8920A/B, 8921A or 8924C.
50 COM /Addr/Hp83236a,Hp892x
60 Hp83236a=719 ! HP-IB Address
70 Hp892x=714 ! HP-IB Address
80 ! Configure test set to display duplex screen.
90 ! If using the 8924C, you must use the command "CONF:RFD 'Freq''
100 OUTPUT Hp892x;"DISP DUPL"
110 OUTPUT Hp892x;"RFG:OUTP 'Dupl''
120 OUTPUT Hp83236a;"RF:PATH 2" ! RF IN/OUT port on HP 83236A.
130 Set_output_freq(1930) ! Frequency in MHz. Always set frequency,
140 ! then level.
150 Set_output_lvl(-40) ! Level from RF IN/OUT of 83236A
160 END
170 Set_output_lvl: SUB Set_output_lvl(Level)
180 COM /Addr/Hp83236,Hp892x
190 Check_gen_cal
200 OUTPUT Hp83236;"RX:TSET:LEV? ";Level;"DBM" ! Set and query
210 ENTER Hp83236;Hp892x_level ! Read the required level for HP 892x
220 OUTPUT Hp892x;"RFG:AMPL ";Hp892x_level;"DBM" ! Set the HP 892x
230 SUBEND
240 Set_output_freq: SUB Set_output_freq(Gen_freq)
250 COM /Addr/Hp83236,Hp892x
260 OUTPUT Hp83236;"RX:TSET:FREQ? ";Gen_freq;"MHZ"
270 ENTER Hp83236;Hp892x_freq
280 ! Units returned from HP 83236 are in Hertz
290 OUTPUT Hp892x;"RFG:FREQ ";Hp892x_freq
300 SUBEND
310 Tcal_gen_lvl: SUB Tcal_gen_lvl
320 COM /Addr/Hp83236,Hp892x
330 ! This routine will perform the temperature compensation
340 ! for all 5 bands of the 83236 generator.
350 !
360 ! Note: This routine modifies the current RF level and freq.
370 !
380 FOR Testband=0 TO 4
390 OUTPUT Hp83236;"COMP:TEMP:FREQ? ";Testband
400 ENTER Hp83236;Test_set_freq
410 OUTPUT Hp83236;"COMP:TEMP:LEV?"
420 ENTER Hp83236;Test_set_level
430 OUTPUT Hp892x;"RFG:FREQ ";Test_set_freq
440 OUTPUT Hp892x;"RFG:AMPL ";Test_set_level
450 WAIT 1 ! Give test set time to set up
460 OUTPUT Hp83236;"COMP:TEMP:EXEC?"
470 ENTER Hp83236;Done_flag
480 IF Done_flag=0 THEN
490 PRINT "Error: Generator Temperature Cal Failed"
PRINT "**** PROGRAM ENDED ****"
STOP
END IF
NEXT Testband
SUBEND

Check_gen_cal: SUB Check_gen_cal
COM /Addr/Hp83236,Hp892x
! This routine checks if the generator calibration is
! required for the current band. The 83236B allows you
! to query if a specific band needs calibration.
! If it is required, this routine stores the current
! RF frequency and calls the calibration routine
! to calibrate the generator. The current RF frequency
! is restored at the end of this routine.
!
! First determine if calibration is required
OUTPUT Hp83236;"COMP:TEMP:REQ:STAT?"
ENTER Hp83236;Gen_cal_req
IF Gen_cal_req=1 THEN
! Save current output frequency
OUTPUT Hp83236;"RX:OUTP:FREQ?"
ENTER Hp83236;Old_gen_freq
! Call the temperature cal routine.
Tcal_gen_lvl
! Reset output frequency
Set_output_freq(Old_gen_freq/1.E+6)
END IF
SUBEND
Using the HP 83236 as an RF Analyzer

Using the HP 83236 as a downconverter for the HP 892X RF analyzer is fairly simple. The two things to consider are level and frequency. To simplify level setting, the HP 83236 has a built-in auto-attenuation circuit that will try to optimize the signal level for the HP 892X. The local oscillator used in the HP 83236A as the downconverter has a fixed frequency of 1050 MHz. The local oscillator used in the HP 83236B has a variable frequency of 970 or 1050 MHz. The routines used in this document will work for either instrument.

Setting the RF Analyzer Frequency

Because the downconvert oscillator is variable, it is necessary to query the HP 83236 to determine the RF analyzer frequency of the HP 892X. By querying the HP 83236, this also programs the HP 83236 so that it will use the correct calibration factors for power measurements and auto-ranging. An example of the commands to perform this function is:

```
OUTPUT 719; "TX:TSET:FREQ? 1930 MHz"
ENTER 719; Hp892x_freq
```

This will let the HP 83236 know the input frequency and allow it to make calibrated power measurements and perform auto-ranging correctly. After sending the command to the HP 83236, the programmer would then set the 892X RF Analyzer frequency. The command for setting the HP 892X to this frequency is:

```
OUTPUT 714; "RFAN:FREQ ";Hp892x_freq
```

Because the auto-ranging circuit uses calibration tables that are a function of frequency. It is important to program the HP 83236 input frequency before selecting the auto-range function or making power measurements.
Setting the HP 83236 frequency as an RF Analyzer

START

Set and Query the HP 83236 TX Input Frequency

OUTPUT Hp83236;"TX:TSET:FREQ? 1930 MHz"

ENTER Hp83236;Hp892x_freq

Set the HP 892X Frequency

OUTPUT Hp892x;"RFAN:FREQ ";Hp892x_freq

DONE
Setting the RF Analyzer Level

The HP 83236 has a 0 to 40 dB programmable input attenuator that can be auto-ranged. This functionality is useful so that over a wide range of input levels, the output level from the HP 83236 into the HP 892X is fairly constant. By default, the auto-range circuit tries to set the output level from the HP 83236 to approximately -9 dBm. For example, if the PCS input signal is at +20 dBm, and the auto-range function is selected on the HP 83236, the input signal will be measured and approximately 29 dB of total attenuation will be used to set the output level from the HP 83236.

The HP-IB commands to force the HP 83236 to auto-range is:

```
OUTPUT 719; "TX:OUTP:LEV:ADJ "
```

Note: The auto-range function is just a one time event and it does not put the HP 83236 into a continuous auto-ranging function. Each time you want the HP 83236 to auto-range, you must send the previous command.

In addition to the auto-range function in the HP 83236, the HP 892X also has auto-ranging of its own input attenuators. For applications where auto-ranging is sufficient, simply leave the HP 892X in auto-ranging (this is continuous) and send the HP-IB command "TX:OUTP:LEV:ADJ " to the HP 83236 whenever the mobile output power is changed. Remember that for accurate auto-ranging, it is necessary to specify the input frequency to the HP 83236 before executing the auto-ranging function. See the section of this document on "Setting the RF Analyzer Frequency".

For more advanced applications, you may want to directly set the programmable attenuator or query the HP 83236 for the total attenuation.

An example of the command to directly set the 0 to 40 dB attenuator to 10 dB is:

```
OUTPUT 719; "TX:OUTP:ATT 10"
```

Remember, this sets the attenuator to 10 dB, but there is additional attenuation due to the splitters and mixers. This additional attenuation is typically 6 to 12 dB depending on frequency.

If you need to query the total attenuation (insertion loss) of the HP 83236 analyzer path, use the following commands:

```
OUTPUT 719; "TX:OUTP:PATH:IL?"
ENTER 719; Insertion_loss
```

This command is especially useful when using the HP 83236 with the HP 8924C.
HP BASIC Example Program for RF Analyzer

10 ! re-save "c:\hpbasic\diet\ANALYZER.PGM"
20 ! Rev. A.02.00
30 ! Example of programming the HP 83236
40 ! for use as a downconverter/analyzer.
50 COM /Addr/Hp83236a,Hp892x
60 Hp83236a=719 ! HP-IB Address
70 Hp892x=714 ! HP-IB Address
80 ! Configure test set to display duplex screen.
90 ! If using 8924C, must also use command "CONF:RFD 'Freq''
100 OUTPUT Hp892x;"DISP DUPL"
110 OUTPUT Hp892x;"RFAN:TMODE 'Manual''
120 OUTPUT Hp892x;"RFAN:INP 'Ant''
130 OUTPUT Hp83236a;"RF:PATH 2 " ! Use RF IN/OUT port on 83236A.
140 Set_input_freq(1930) ! Frequency in MHz. Always set frequency,
150 ! then level.
160 Auto_range ! Auto-range 83236A input
170 Set_attenuator(15) ! Set attenuator to 15 dB
180 Insertion_loss(1930) ! Read the input path insertion loss at
190 ! 1930 MHz.
200 END
210 Set_input_freq: SUB Set_input_freq(Freq)
220 COM /Addr/Hp83236,Hp892x
230 OUTPUT Hp83236;"TX:TSET:FREQ? ";Freq;"MHZ"
240 ENTER Hp83236;Hp892x_freq
250 OUTPUT Hp892x;"RFAN:FREQ ";Hp892x_freq
260 SUBEND
270 Auto_range: SUB Auto_range
280 COM /Addr/Hp83236,Hp892x
290 ! Note: For this to work, RF Path must be
300 ! set to 1 or 2.
310 OUTPUT Hp83236;"TX:OUTPUT:LEV:ADJ "
320 SUBEND
330 Set_attenuator: SUB Set_attenuator(Atten_value)
340 COM /Addr/Hp83236,Hp892x
350 OUTPUT Hp83236;"TX:OUTP:ATT ";Atten_value;" dB"
360 SUBEND
370 Insertion_loss: SUB Insertion_loss(Freq)
380 COM /Addr/Hp83236,Hp892x
390 OUTPUT Hp83236;"TX:OUTP:PATH:IL? ";Freq;" MHz"
400 ENTER Hp83236;Loss
410 PRINT "For frequency ";Freq;" MHz"
420 PRINT "Path Loss is ";Loss
430 SUBEND
Making Accurate Power Measurements

Calibration

Like most other power meters, the HP 83236 power meter must occasionally be zeroed to make accurate measurements. This process takes <1 second and should be performed whenever there is a temperature change in the HP 83236.

The HP-IB command to zero the power meter is:

```
OUTPUT 719; "COMP:PDET:DCOF:EXEC"
```

In many power meters, it is necessary to remove all input signals before zeroing. This is not necessary with the HP 83236. Whenever the command is sent to zero the power meter, the internal circuitry in the HP 83236 has an automatic switch which temporarily disconnects the power meter input, performs the zeroing, and then reconnects the input.

In addition to the programmer specifically performing the zeroing function, there are two other ways that the power meter is automatically zeroed. Whenever the instrument is powered up or reset ("*RST"), the power meter is automatically zeroed. Also, whenever the temperature calibration for the RF generator is performed ("COMP:TEMP:EXEC?") this causes the power meter to be zeroed. In many applications, the RF generator calibration is checked and performed quite often and it isn't necessary that the programmer specifically perform the power meter zeroing.

Making Measurements

When making RF power measurements, there are four factors that need to be considered to determine how the measurement is performed. Your application will determine the optimum setting for each of these factors. The HP 83236 operating manual (Chapter 3) describes these recommended settings in great detail. The four factors to consider are triggering, samples to ignore, samples to collect, and averaging.
**Triggering**
For CW or constant amplitude signals, the recommended triggering mode is "Immediate". The HP-IB command for immediate triggering is:

```
OUTPUT 719;"TX:INP:POW:TRIG IMM"
```

For pulsed signals like TDMA mobiles, the recommended triggering mode is "Amplitude" triggering. The HP-IB command for amplitude triggering is:

```
OUTPUT 719;"TX:INP:POW:TRIG AMPTD"
```

For low level, pulsed signals < -5 dBm, it is necessary to use an external triggering signal which is input to the HP 83236. To select "External" triggering, use the HP-IB command:

```
OUTPUT 719;"TX:INP:POW:TRIG EXT"
```

**Samples to ignore**
For pulsed signals, you may want to ignore the first few samples of the measurement. This is because the signal is being pulsed and you will want to measure the pulse power when it is fully on, not during the rise time. For TDMA mobiles using amplitude triggering, the recommended number of samples to ignore is 17. To program the detector for this operation, use the HP-IB command:

```
OUTPUT 719;"TX:INP:POW:PDET:SAMP:IGN 17"
```

**Samples to collect**
The internal measurement circuits make measurements at a rate of one measurement each 12.5 microseconds. For accurate measurements, you will want to collect a large number of measurements. For TDMA mobiles, the recommended number of measurements are 461. For CDMA signals, the recommended measurement samples are 4800. To program the detector to use 4800 samples for the power measurement, use the HP-IB command:

```
OUTPUT 719;"TX:INP:POW:PDET:SAMP:LENG 4800"
```

**Averaging**
In addition to the number of samples, you may also want to use averaging to improve measurement accuracy. The HP 83236 operating manual contains formulas you can use to calculate the accuracy with averaging. To program the detector to use five averages for power measurements, use the HP-IB command:

```
OUTPUT 719;"TX:INP:POW:PDET:SAMP:AVER 5"
```
Using the HP 83236 as an RF Power Meter

START

Zero Power Meter

OUTPUT Hp83236;"COMP:PDET:DCOF:EXEC"

Set up a Phone Call

OUTPUT Hp83236;"TX:INP:POW:TRIG IMM"

Configure:
* Trigger Type
* Ignore Samples
* Sample Length
* Number of Averages

OUTPUT Hp83236;"TX:INP:POW:PDET:SAMP:IGN 3"
OUTPUT Hp83236;"TX:INP:POW:PDET:SAMP:LENG 500"
OUTPUT Hp83236;"TX:INP:POW:PDET:SAMP:AVG 1"

Make the Measurement

OUTPUT Hp83236;"TX:INP:POW?"
ENTER Hp83236; Rf_power

DONE
HP BASIC Example Program for Using the Power Meter

10 ! re-save "c:\ hpbasic\ diet\ POWER.PGM"
20 ! Rev. A.02.00
30 ! Example program for measuring power
40 ! with the HP 83236.
50 COM /Addr/Hp83236a,Hp892x
60 Hp83236a=719 ! HP-IB Address
70 Hp892x=714 ! HP-IB Address
80 ! Configure test set to display duplex screen.
90 OUTPUT Hp892x;"DISP DUPL"
100 OUTPUT Hp892x;"RFAN:TMODE 'Manual'"
110 OUTPUT Hp892x;"RFAN:INP 'Ant'"
120 OUTPUT Hp83236a;"RF:PATH 2 " ! Use RF IN/OUT port on 83236A.
130 Set_input_freq(1930) ! Frequency in MHz.
140 Zero_power ! Zero the power meter.
150 Measure_power("CDMA") ! Measure power
160 ! 1930 MHz.
170 END
180 Set_input_freq: SUB Set_input_freq(Freq)
190 COM /Addr/Hp83236,Hp892x
200 OUTPUT Hp83236;"TX:TSET:FREQ? ";Freq;"MHZ"
210 ENTER Hp83236;Hp892x_freq
220 OUTPUT Hp892x;"RFAN:FREQ ";Hp892x_freq
230 SUBEND
240 Zero_power: SUB Zero_power
250 COM /Addr/Hp83236,Hp892x
260 ! This will temporarily disconnect the
270 ! input signal. It should be performed
280 ! before a voice channel is established.
290 OUTPUT Hp83236;"COMP:PDET:DCOF:EXEC"
300 SUBEND
310 Measure_power: SUB Measure_power(System_type$)
320 COM /Addr/Hp83236,Hp892x
330 SELECT System_type$
340 CASE "CDMA"
350 Ignore_samples=3
360 Sample_length=4800
370 Averages=1
380 Trig_type$="IMM"
390 CASE "TDMA BASE"
400 Ignore_samples=3
410 Sample_length=1000
420 Averages=1
430 Trig_type$="IMM"
440 CASE "TDMA MOBILE HIGH POWER"
450 Ignore_samples=17
460 Sample_length=461
470 Averages=3
480 Trig_type$="AMPTD"
490 CASE "TDMA MOBILE LOW POWER"
Slot_number=1

! Calculate ignore value base on slot number.
Ignore_samples=17+(Slot_number-1)*533
Sample_length=461
Averages=3
Trig_type$="EXT"

END SELECT

! Initialize detector characteristics
OUTPUT Hp83236;"TX:INP:POW:TRIG ";Trig_type$
OUTPUT Hp83236;"TX:INP:POW:PDET:SAMP:IGN ";Ignore_samples
OUTPUT Hp83236;"TX:INP:POW:PDET:SAMP:LENG ";Sample_length
OUTPUT Hp83236;"TX:INP:POW:PDET:SAMP:AVER ";Averages

! Make the measurement
OUTPUT Hp83236;"TX:INP:POW?"
ENTER Hp83236;Rf_power
IF Rf_power=-130 THEN
PRINT "Power Measurement invalid"
STOP
END IF
IF Rf_power=-201 THEN
PRINT "Trigger not detected"
STOP
END IF
PRINT "Measured power is ";Rf_power

SUBEND